

第九章 螢幕控制

設定顯示模式
設定游標大小和位置
滾動螢幕
顯示字和屬性
繪圖模式操作

本章介紹的系統呼叫 (1)

INT 10H

Function 00H: 設定顯示模式

Function 01H: 設游標大小

Function 02H: 設游標位置

Function 03H: 讀游標狀況

Function 05H: 選擇顯示頁

Function 06H: 向上滾動螢幕

Function 07H: 向下滾動螢幕

Function 08H: 讀字和屬性

Function 09H: 顯示字和屬性

Function 0AH: 顯示字

本章介紹的系統呼叫 (2)

Function 0BH: 設定調色盤


Function 0CH: 寫一影像點 (pixel, picture element)

Function 0DH: 讀一影像點


Function 0FH: 讀顯示模式

Function 13H: 顯示字串

螢幕掃描方式




Noninterlaced scan




Interlaced scan

文字顯示



控制電路



顯示模式

模式	列*行	型式	顯示 位置	頁數	解析度	顏色
00	25*40	彩色	B800	0-7	360*400	16
01	25*40	彩色	B800	0-7	360*400	16
02	25*80	彩色	B800	0-3	720*400	16
03	25*80	彩色	B800	0-3	720*400	16
07	25*80	單色	B000	0	720*400	

INT 10H Function 00H: 設定顯示模式

```
MOV    AH, 0
MOV    AL, mode
INT    10H
```

設定模式, 並清螢幕

若不想清螢幕, 應設 mode 的位元 7 為 1

INT 10H Function 01H: 設游標大小

CH的低階四位元(位元5-0)放游標開始列號
CL的低階四位元放游標結束列號

```
MOV     AH, 1
MOV     CH, 5
MOV     CL, 6
INT     10H
```

0	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>
1	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>
2	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>
3	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>
4	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>
5	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>
6	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>
7	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>
8	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>
9	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>
10	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>
11	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>
12	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>
13	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>
14	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>
15	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>

INT 10H Function 03H: 讀游標狀況

```
MOV     AH, 3
MOV     BH, page#
INT     10H
```

送回

CH: 游標開始列號, CL: 游標結束列號

DH: 游標位置, 列號, DL: 游標位置, 行號

INT 10H Function 05H:
選擇顯示頁

```
MOV    AH, 5
MOV    AL, page#
INT    10H
```

INT 10H Function 08H:
讀字和屬性

```
MOV    AH, 8
MOV    BH, page#
INT    10H
```

送回

游標位置顯示字的

AH: 屬性, AL: ASCII碼

INT 10H Function 09H:
顯示字和屬性

```
MOV     AH,9
MOV     AL,character_code
MOV     BH,page#
MOV     BL,attribute
MOV     CX,repeat_count
INT     10H
```

INT 10H Function 0AH:
顯示字

```
MOV     AH,0AH
MOV     AL,character_code
MOV     BH,page#
MOV     CX,repeat_count
INT     10H
```

使用原來的屬性

INT 10H Function 0FH: 讀顯示模式

```
MOV     AH,0FH  
INT     10H
```

送回

AL:顯示模式, AH: 行數, BH: 目前頁號

INT 10H Function 13H 顯示字串


```
MOV AH,13H  
MOV AL,subfunction  
MOV BH,page#  
MOV BL,attribute  
LEA BP,address  
MOV CX,length  
MOV DH,row  
MOV DL,column  
INT 10H
```


INT 10H Function 13H Subfunctions

- 00: 顯示字串和屬性，游標不動
- 01: 顯示字串和屬性，前進游標
- 02: 顯示字串和接著的屬性，游標不動
- 03: 顯示字串和接著的屬性，前進游標

例子 1: 顯示ASCII文字集合

儲存原顯示模式, 結束前恢復
設顯示模式為 3
清螢幕, 以當時游標位置顏色設螢幕顏色
在螢幕中央以 16*16 方陣方式顯示 256 個
ASCII 文字



主程式(1)

```
.model    small
.stack   64
.data
ctr db    0      ;counter for ascii characters
col db    24     ;column of screen
row db    4      ;row of screen
mode db   ?      ;video mode
.286
;-----
.code
a10main proc    near
    mov ax,@data ;initialize segment registers
    mov ds,ax
    mov es,ax
    call b10mode ;get/set video mode
    call c10clr  ;clear screen
```

主程式(2)

```
a20:
    call d10set      ;set cursor
    call e10disp    ;display characters
    cmp  ctr,0ffh   ;last character displayed?
    je   a30        ;yes, exit
    inc  ctr        ;increment char. counter
    add  col,2      ;increment column
    cmp  col,56     ;at end of column?
    jne  a20        ; no, bypass
    inc  row        ; yes, increment row
    mov  col,24     ; and reset column
    jmp  a20
```

主程式(3)

```
a30:
    mov  ah,10h     ;request get character
    int  16h
    mov  ah,0       ;request set mode
    mov  al,mode
    int  10h
    mov  ax,4c00h   ;end
    int  21h       ; processing
a10main endp
```

儲存原顯示模式, 設定新模式

```
; get and set video mode:
; -----
b10mode proc near
    mov ah,0fh          ;request get mode
    int 10h
    mov mode,al        ;save mode
    mov ah,0           ;request set new mode
    mov al,3           ;standard color
    int 10h
    ret
b10mode endp
```

清螢幕

```
; clear screen and create window:
; -----
c10clr    proc    near
    pusha                ;preserve general registers
    mov ah,8             ;request get current
    int 10h              ; attribute in ah
    mov bh,ah            ;move it to bh
    mov ax,0600h         ;scroll full screen
    mov cx,0             ;upper left location
    mov dx,184fh         ;lower right location
    int 10h              ;call bios
    mov ax,0610h         ;create 16-line window
    mov bh,16h           ;blue back, brown foreground
    mov cx,0418h         ;upper left corner 04:24
    mov dx,1336h         ;lower right corner 19:54
    int 10h
    popa                 ;restore general registers
    ret                  ;return to caller
c10clr    endp
```

設游標位置

```
; set cursor to row and column:
; -----
d10set    proc near
    pusha                ;preserve general registers
    mov  ah,2            ;request set cursor
    mov  bh,0            ;page number 0
    mov  dh,row          ;new row
    mov  dl,col          ;new column
    int  10h            ;call bios
    popa                ;restore general registers
    ret
d10set    endp
```

顯示字

```
; display ASCII characters:
; -----
e10disp  proc    near
    pusha                ;preserve general registers
    mov  ah,0ah          ;request display
    mov  al,ctr          ;ascii character
    mov  bh,0            ;page 0
    mov  cx,1            ;one character
    int  10h
    popa                ;restore general registers
    ret                  ;return to caller
e10disp  endp
        end  a10main
```

例子 2

反白(白底藍字)顯示提示字 “Name?”

輸入名字(藍底白字)

右邊對齊70行反白顯示名字

資料定義

```
.model    small
.stack   1024
;-----
.data
parlist  label  byte           ;name parameter list
maxlen   db    20             ; maximum length of name
actlen   db    ?              ; no. of chars entered
kbname   db    20 dup(' ')    ; name
;
left_col  equ   51            ;left column for display
bott_scrn equ   23           ;bottom row for display
attrib   db    0              ;screen attribute
col       db    5              ;screen column
row       db    0              ; and row
prompt   db    'Name? '      ;prompt for input
```

主程式(1)

```
.code
; main procedure:
; -----
a10main proc    far
    mov  ax,@data    ;initialize segment
    mov  ds,ax       ; registers
    mov  es,ax
    mov  ax,1003h    ;disable blink
    mov  bl,0        ; attribute
    int  10h
    mov  al,00h      ;clear screen
    call q10scr1
```

主程式(2)

```
a20:
    mov  col,05      ;set column to 05
    call b10prom     ;display prompt
    call c10inpt     ;provide for input of name
    cmp  actlen,0    ;no name?
    je   a30         ; yes, exit
    call d10name     ;display name
    jmp  a20

a30:
    mov  al,00h      ;exit,
    call q10scr1     ;clear screen
    mov  ax,4c00h    ;end of processing
    int  21h
a10main endp
```

顯示提示字

```
; display prompt:
; -----
b10prom proc    near
    mov  attrib,71h;set attribute
    lea  bp,prompt ;set address of prompt
    mov  cx,6      ; and length
    call e10disp  ;display routine
    ret
b10prom endp
```

輸入名字

```
; accept input of name from keyboard:
; -----
c10inpt proc    near
    mov  ah,0ah      ;request keyboard
    lea  dx,parlist
    int  21h
    ret
c10inpt endp
```


顯示名字

```
; set up for displaying name, scroll if
; near bottom of the screen:
; -----
d10name proc      near
    pusha          ;preserve general registers
    mov    al,maxlen ;calculate screen indent,
    sub    al,actlen
    add    al,left_col ; add to left column,
    mov    col,al    ; and store
    mov    attrib,17h ;reverse video
    lea    bp,kbname ;initialize name
    mov    cx,actlen ; and length
    call   e10disp   ;display name
    cmp    row,bott_scrn ;near bottom of screen?
    jae    d30
    inc    row        ; no, increment row
    jmp    d90        ; and exit
d30:
    mov    al,1      ;
    call   q10scrl   ; scroll screen
d90:
    popa           ;restore general registers
    ret
d10name endp
```

顯示一個字和屬性

```
; display character and set attribute:
; -----
e10disp proc      near ;BP, CX set on entry
    pusha          ;preserve general registers
    mov    ah,13H   ;request display
    mov    al,1     ; characters
    mov    bh,0     ;page number
    mov    bl,attrib ;attribute
    mov    dh,row   ;screen row
    mov    dl,col   ; and column
    int    10h
    popa           ;restore general registers
    ret           ;return to caller
e10disp endp
```

清螢幕

```
; scroll screen and set attribute:
; -----
q10scrl proc    near
    pusha                ;preserev general registers
    mov  ah,6            ;request scroll screen
    mov  bh,86h          ;grey, brown
    mov  cx,0            ;full screen
    mov  dx,184fh
    int  10h            ;call bios
    popa                ;restore general registers
    ret
q10scrl endp
    end  a10main
```

繪圖顯示模式

模式	型式	位置	頁數	解析度	顏色
04H	彩色	B800	8	320*200	4
05H	彩色	B800	8	320*200	4
06H	彩色	B800	8	640*200	2
0DH	彩色	B800	8	320*200	16
0EH	彩色	A000	4	640*200	16
0FH	單色	A000	2	640*350	1
10H	彩色	A000	2	640*350	16
11H	彩色	A000	1	640*480	2
12H	彩色	A000	1	640*480	16
13H	彩色	A000	1	320*200	256

INT 10H Function 0BH 設定調色盤

- BH=0, 設背景顏色(放在BL)

```
MOV  AH,0BH  
MOV  BH,0  
MOV  BL,4  
INT  10H
```

- BH=1, 選擇調色盤(BL=0,1)

```
MOV  AH,0BH  
MOV  BH,1  
MOV  BL,0  
INT  10H
```

INT 10H Function 0CH 寫一影像點

AL:顏色, CX:Column, BH:頁號, DX: Row

```
MOV  AH,0CH  
MOV  AL,3  
MOV  BH,0  
MOV  CX,200  
MOV  DX,50  
INT  10H
```

INT 10H, Function 10H, Subfn 0 設定調色盤顏色

```
MOV  AH,10H
MOV  BH,02    ;顏色
MOV  BL,01    ;調色盤值
INT  10H
```

繪圖模式的例子: 顯示文字

- 繪圖模式下顯示文字
- 以INT 10H function 13H顯示字串
- 接著以function 0AH重覆顯示笑臉10次

```

.model small
.stack 64
.data
string db '1234567890'
.286
;-----
.code
a10main proc far
mov ax,@data ;establish
mov ds,ax ;addressability
mov es,ax
mov ah,0fh ;get
int 10h ; original
push ax ; video mode and save
call b10mode ;set video mode
call c10scrl ;scroll screen
call d10strg ;function 13h to display
call e10disp ;function 0ah to display
mov ah,10h ;request keyboard
int 16h ; input
pop ax
mov ah,0 ;restore
int 10h ; original mode (in AL)
mov ax,4c00h ;end
int 21h ; processing
a10main endp

```

```

; set graphics mode and request palette:
; -----
b10mode proc near ;uses AV and BX
mov ah,0 ;set EGA/VGA graphics mode
mov al,12h ;640 cols x 480 rows
int 10h
mov ah,0bh ;set background palette
mov bh,0 ;background
mov bl,07h ;grey
int 10h
ret
b10mode endp

```

```

c10scri proc    near
    pusha      ;preserve general registers
    mov       ax,0605h    ;request scroll 5 rows
    mov       bh,1110b    ;yellow color
    mov       cx,0        ;from row:column
    mov       dx,044fh    ;to row:column
    int       10h
    popa      ;restore registers
    ret       ; yes, terminate
c10scri endp

```

```

;      displaying string, set attribute and cursor:
d10strg proc    near
    pusha      ;preserve general registers
    mov       ax,1301h    ;request display
    mov       bx,21h      ;page:attribute
    lea      bp,string    ;character string
    mov       cx,10       ;length
    mov       dx,0815h    ;row:column
    int       10h
    popa      ;restore registers
    ret
d10strg endp

```

```
;      display character repeatedly
e10disp proc    near
    pusha      ;preserve general registers
    mov       ax,0a01h ;request display
    mov       bh,0     ;happy face
    mov       bl,0100b ;red color
    mov       cx,10    ;10 times
    int       10h
    popa      ;restore registers
    ret
e10disp endp
end        a10main
```

繪圖模式的例子
顯示圖形

```

        .model small
        .stack 64

.286
;-----
        .code
a10main proc near
        mov     ax,@data       ;establish
        mov     ds,ax         ; addressability
        mov     es,ax
        mov     ah,0fh        ;get
        int     10h           ; original
        push    ax            ; video mode and save
        call    b10mode       ;set video mode
        call    c10disp       ;display color graphics
        mov     ah,10h        ;request keyboard
        int     16h           ;response
        pop     ax
        mov     ah,0          ;restore
        int     10h           ; original mode (in AL)
        mov     ax,4c00h      ;end
        int     21h           ; processing
a10main endp

```

```

;   set graphics mode and palette:
;   -----
b10mode proc    near
        mov     ax,012h       ;request graphics mode
        int     10h           ;640 cols x 480 rows
        mov     ah,0bh        ;set background palette
        mov     bx,07h        ;background grey
        int     10h
        ret
b10mode endp

```



```

;      display 210 rows of graphics dots, 512
;      columns, change color for each row:
;      -----
c10disp proc      near
    pusha                ;preserve general registers
    mov     bx,0          ;set initial page,
    mov     cx,64         ; color, column,
    mov     dx,70         ; and row
c20:
    mov     ah,0ch        ;write pixel dot
    mov     al,b1         ;set color
    int     10h          ;bx, cx, and dx are
preserved
    inc     cx            ;increment column
    cmp     cx,576        ;column at 576?
    jne     c20           ; no, loop
    mov     cx,64         ; yes, reset column
    inc     bl            ;change color
    inc     dx            ;increment row
    cmp     dx,280        ;row at 280?
    jne     c20           ; no, loop
    popa                 ;restore registers
    ret                  ; yes, terminate
c10disp endp
end      a10main

```

直接寫資料到顯示記憶體

- Mode 3 page 1 的地址由B900[0]開始
- 一個字佔2個位元組: ASCII碼和屬性
- 由第5列第10字起, 每列顯示一個字母60次
- 字母由A起, 到P止

```

        .model small
        .stack 64
video_seg segment at 0b900h ;page 1 of video area
vid_area db 1000h dup (?)
video_seg ends
        .286
;-----
        .code
a10main proc far
        mov ax,video_seg ;addressibility for
        mov es,ax ; video area
        assume es:video_seg
        mov ah,0fh ;request get
        int 10h ; and save
        push ax ; current mode
        push bx ; and page
        mov ax,3 ;set mode 3,
        int 10h ; clear screen
        mov ax,0501h ;set page #01
        int 10h
        call b10disp ;process display area

```

```

        mov ah,10h ;wait for keyboard
        int 16h ; response
        mov ah,5 ;restore original
        pop bx ;page number
        mov al,bh
        int 10h
        pop ax ;restore video mode
        mov ah,0 ;
        int 10h
        mov ax,4c00h ;end of processing
        int 21h
a10main endp

```

```

b10disp    proc    near
    pusha                    ;preserve general registers
    mov     al,41h           ;character to display
    mov     ah,1             ;attribute
    mov     di,820           ;start of display area
b20:
    mov     cx,60            ;characters per row
b30:
    mov     es:word ptr[di],ax ;character to display
    add     di,2             ;next attribute + character
    loop    b30              ;repeat 60 times
    inc     ah               ;next attribute
    inc     al               ; and character
    add     di,40            ;indent for next row
    cmp     al,51h           ;last character to display?
    jne     b20              ; no, repeat
    popa                     ;restore registers
    ret
b10disp    endp
end        a10main

```

畫表格框架的文字碼

水平線	C4H	CDH
垂直線	B3H	BAH
左上角	DAH	C9H
左下角	C0H	C8H
右上角	BFH	BBH
右下角	D9H	BCH
左中線	C3H	CCH
右中線	B4H	B9H
上中線	C2H	CBH
下中線	C1H	CAH
十字	C5H	CEH

一個 Menu 的定義

```
MENU      DB      0DAH, 17 DUP (0C4H), 0BFH
          DB      0B3H, ` Add records      `, 0B3H
          DB      0B3H, ` Delete records   `, 0B3H
          DB      0B3H, ` Enter orders     `, 0B3H
          DB      0B3H, ` Print report     `, 0B3H
          DB      0B3H, ` Update accounts  `, 0B3H
          DB      0B3H, ` View records    `, 0B3H
          DB      0C0H, 17 DUP (0C4H), 0D9H
```